Magnetic Field Created by a Distribution of Conductor

Emmanuel Froidefond

Laboratoire de Physique Subatomique et de Cosmologie
53 avenue des Martyrs
38026 GRENOBLE CEDEX
FRANCE
froidefond@lpsc.in2p3.fr

Bruno Autin

CERN
Geneva
Bruno.Autin@cern.ch

The magnetic field created by a distribution of conductors can be calculated by a simple Mathematica function which applies a general theory to a large variety of practical cases. The magnetic field is proportional to the current and given by Ampère's law for an infinitely long conductor. This assumption reduces the problem to a 2D calculation.

A unit current in a point conductor j produces a horizontal or vertical field component aij at the observation point i. The components aij are the coefficients of a matrix A. The wanted field distribution is described by a vector b of components bi and the unknown currents by a vector x of components xj. The number of observation points is much larger than the number of conductors for a stable solution of the system A x + b = 0. The solution is given using either the generalized inverse of A or an iterative least squares method when conductors are suspected not to be independent.

The *Mathematica* function is

ConductorField[obs, cur1,cur2,...,opts]

with

obs, vector of observation points

curj, vector of current points in sheet j (j=1,2,...)

opts the two possible options:

FieldComponent −> Horizontal or Vertical (default) for horizontal or vertical field components.

Iteration −> True for an iterative solution or False (default) when the generalized inverse is invoked.

The sequence of current sheets is made of real conductors and of their images in the iron shield. The function is used to determine the initial distribution of currents. Then the distribution and the geometry of the shield are entered as input to a numerical code which evaluates the effects due to edges and finite iron permeability. Last the function is used again to correct the errors revealed by the numerical code.

■ How to produce a radial magnetic field suiting $B_0\left(\frac{x}{x_0}\right)^k$



Why produce such a field? The old method, but recently reintroduced, of particle acceleration with Fixed Field Alternating Gradient synchrotrons (FFAG) needs it. Dipoles with such field, distributed along a ring, allow particles to be accelerated on a spiral trajectory with a relatively small variation of the orbit radius. One advantage for this kind of accelerators is to avoid radio frequency variation of the magnetic field. The first way this field was produced was a pole shaped dipole (as shown below). But the variation of the field is fixed (k is left constant), and there is no way to apply small corrections to the field.



The second way to produce this field is to design a dipole with flat poles, and add series of conductors to the poles. Each conductor have its own current intensity.



The picture above is the result of a *Poisson* calculation. The yoke is the zone where field lines are concentrated. The small squares are the conductors. The vertical field component profile along the ring radius is shown in the graph below.



■ Definition of the problem

□ General description

We first consider one conductor. Let the conductor coordinates be $\{\xi,\eta\}$, and the observation point coordinates be $\{x, 0\}$. From Biot and Savart law, the components of the field are then :

– vertical : $B_y(x, \xi, \eta)=\frac{\mu_0\ I}{4\pi}\ \frac{x-\xi}{(x-\xi)^2+\eta^2}$

– horizontal : $B_x(x, \xi, \eta)=\frac{\mu_0\ I}{4\pi}\ \frac{\eta}{(x-\xi)^2+\eta^2}$

The horizontal component is the one to be compared to the searched field $B_0\left(\frac{x}{x_0}\right)^k$. If we consider several conductors, we now have :

$B_y(x,\{\xi_i,\eta_i\})=\frac{\mu_0}{4\pi}\sum_{i=1}^{N} I_i\ \frac{x-\xi_i}{(x-\xi_i)^2+\eta_i{}^2}$

where N is the number of conductors. If we now consider a vector of observation points, we just to put an index j to the coordinate x.

◻ Matrix description

The problem we want to solve is the following : how can we calculate the intensity of each conductor, knowing the field to obtain? We can treat this as a matrix problem where, if the number of conductors is N and the number of observation points is P, we have :

– a vector of the vertical component of the magnetic field of P elements :

$$B_y = \begin{pmatrix} B_{y1} \\ B_{y2} \\ \ldots \\ B_{yP} \end{pmatrix}$$

– a vector of the intensities in the conductors of N elements :

$$I = \begin{pmatrix} I_1 \\ I_2 \\ \ldots \\ I_N \end{pmatrix}$$

– a matrix of the vertical field expressions for a current intensity of $2\pi/\mu_0$ conductors and observation points of dimensions (PxN) :

$$A = \begin{pmatrix} \frac{x_1-\xi_1}{(x_1-\xi_1)^2+(y_1-\eta_1)^2} & \ldots & \frac{x_1-\xi_N}{(x_1-\xi_N)^2+(y_1-\eta_N)^2} \\ \ldots & \ldots & \ldots \\ \frac{x_P-\xi_1}{(x_P-\xi_1)^2+(y_P-\eta_1)^2} & \ldots & \frac{x_P-\xi_N}{(x_P-\xi_N)^2+(y_P-\eta_N)^2} \end{pmatrix}$$

The system can then be reduced to the simple equation $A \times I + B_y = 0$ that allow the use of the *Mathematica* function `LinearSolve`.

■ The function `ConductorField`

As we have seen above, the calculation of the field can be considered in two ways : one can use symbolic calculation, or matrices formalism. The two methods are introduced in the function `ConductorField` in order to build a function as general as possible. It allow to extract one component (vertical/horizontal) or both components of the field by setting the option variable `FieldComponent` to vertical, horizontal or all.

◻ Symbolic calculation

To introduce symbolic calculation of the magnetic field, the only formula written in the function correspond to the field produced by one conductor. The observation points are always situated in a plane at a distance g from the conductors. That implies the function only needs the *x* variable for the observation point location.

The position of conductors $\{\xi, \eta\}$ is given as a vector. It is built as a list of one element $\{\xi, \eta\}$, or N elements $\{\{\xi_1, \eta_1\}, \{\xi_2, \eta_2\}, \ldots\}$.

```
In[1]:= ConductorField[x_Symbol, cur_?VectorQ, opt___Rule] :=
 Module[{a, fc, g, r2},
  fc = FieldComponent /. {opt} /. Options[ConductorField];
  {a, g} = cur;
  r2 = (a - x) ^2 + g^2;
  Which[fc === Vertical, (a - x) / r2,
   fc === Horizontal, g / r2,
   fc === All, {a - x, g} / r2]
 ]
```

Let the current vector be $\{\xi, \eta\}$, the function `ConductorField` then returns :

Out[16]= $$\left\{\frac{-x+\xi}{\eta^2+(-x+\xi)^2}, \frac{\eta}{\eta^2+(-x+\xi)^2}\right\}$$

Now let the current vector be $\{\{\xi_1,\eta_1\},\{\xi_2,\eta_2\},\{\xi_3,\eta_3\}\}$, then we get the sum of vertical components and of horizontal components :

Out[19]= $$\left\{\frac{-x+\xi_1}{\eta_1^2+(-x+\xi_1)^2}+\frac{-x+\xi_2}{\eta_2^2+(-x+\xi_2)^2}+\frac{-x+\xi_3}{\eta_3^2+(-x+\xi_3)^2}, \frac{\eta_1}{\eta_1^2+(-x+\xi_1)^2}+\frac{\eta_2}{\eta_2^2+(-x+\xi_2)^2}+\frac{\eta_3}{\eta_3^2+(-x+\xi_3)^2}\right\}$$

□ Matrices formalism

The matrix A contains elements of the form :
$\frac{x_j-\xi_i}{(x_j-\xi_i)^2+(y_j-\eta_i)^2}$ or $\frac{y_j-\eta_i}{(x_j-\xi_i)^2+(y_j-\eta_i)^2}$,
where $\{x_j, y_j\}$ are the observation point coordinates, and $\{\xi_i, \eta_i\}$ are the conductor coordinates.
The difference with the symbolic calculation concerns observation points coordinates. They can be given either as a vector or as a matrix, which means that an array of observation points can be studied.

```
ConductorField[obs_?MatrixQ, cur_?VectorQ, opt___Rule] :=
 Module[{dr, du, fc, m, normdr},
  fc = FieldComponent /. {opt} /. Options[ConductorField];
  du = Transpose[obs] - cur;
  dr = Transpose@du;
  normdr = #.# & /@ dr;
  m = Reverse /@ dr / normdr;
  {mh, mv} = Transpose[m];
  Which[fc === Vertical, mv,
   fc === Horizontal, mh,
   fc === All, m]
 ]
```

For one conductor, and a series of 10 observation points in the same plane, the function returns a matrix containing the values of the field components :

In[22]:=
```
x = Range[11] - 6;
y = Table[0, {n, 0, 10}];
ob = Transpose[{x, y}];
cu = {0, 1};
Transpose@ConductorField[ob, cu, FieldComponent → All]
```

$$\begin{pmatrix} -\frac{1}{26} & -\frac{1}{17} & -\frac{1}{10} & -\frac{1}{5} & -\frac{1}{2} & -1 & -\frac{1}{2} & -\frac{1}{5} & -\frac{1}{10} & -\frac{1}{17} & -\frac{1}{26} \\ -\frac{5}{26} & -\frac{4}{17} & -\frac{3}{10} & -\frac{2}{5} & -\frac{1}{2} & 0 & \frac{1}{2} & \frac{2}{5} & \frac{3}{10} & \frac{4}{17} & \frac{5}{26} \end{pmatrix}$$

If we now have two conductors, symmetrically located compared to the plane of the observation points, we get :

```
x = Range[11] - 6;
y = Table[0, {n, 0, 10}];
ob = Transpose[{x, y}];
cu = {{0, 1}, {0, -1}};
Transpose@ConductorField[ob, cu, FieldComponent → All]
```

$$\begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ -\frac{5}{13} & -\frac{8}{17} & -\frac{3}{5} & -\frac{4}{5} & -1 & 0 & 1 & \frac{4}{5} & \frac{3}{5} & \frac{8}{17} & \frac{5}{13} \end{pmatrix}$$

□ Summary

`ConductorField`[x, $\{\xi, \eta\}$, option] : field at location $\{x, 0\}$ produced by a conductor at $\{\xi, \eta\}$with the current intensity $2\pi/\mu_0$.

`ConductorField`[x, $\{\{\xi_1 ,\eta_1 \},\{\xi_2 ,\eta_2 \},...\}$, option] : field in $\{x, 0\}$ produced by conductors at $\{\{\xi_1 ,\eta_1 \},\{\xi_2 ,\eta_2 \},...\}$ with current intensities $2\pi/\mu_0$.

`ConductorField`[$\{\{x_1 ,y_1 \},\{x_2 ,y_2 \},...\}$, $\{\xi, \eta\}$, option] : field at observation points $\{\{x_1 ,y_1 \},\{x_2 ,y_2 \},...\}$ produced by a conductor located at $\{\xi, \eta\}$with the current intensity $2\pi/\mu_0$.

`ConductorField`[$\{\{x_1 ,y_1 \},\{x_2 ,y_2 \},...\}$, $\{\{\xi_1 ,\eta_1 \},\{\xi_2 ,\eta_2 \},...\}$, option] : field produced at observation points $\{\{x_1 ,y_1 \},\{x_2 ,y_2 \},...\}$ by conductors located at $\{\{\xi_1 ,\eta_1 \},\{\xi_2 ,\eta_2 \}$ with current intensities $2\pi/\mu_0$.

Options :
`FieldComponent` → *Vertical*. Vertical field component is calculated (default).
`FieldComponent` → *Horizontal*. Horizontal field component is calculated.
`FieldComponent` → *All*. Both vertical and horizontal field components are calculated.

Finally the total function is given by the following program :

```
In[1]:= Clear[ConductorField]
        Options[ConductorField] = {FieldComponent → Vertical}

        ConductorField[x_Symbol, cur_?VectorQ, opt___Rule] :=
         Module[{a, fc, g, r2},
          fc = FieldComponent /. {opt} /. Options[ConductorField];
          {a, g} = cur;
          r2 = (a - x) ^2 + g^2;
          Which[fc === Vertical, (a - x) / r2,
           fc === Horizontal, g / r2,
           fc === All, {a - x, g} / r2]
         ]

        ConductorField[x_Symbol, cur_?MatrixQ, opt___Rule] :=
         Plus @@ (ConductorField[x, #, opt] & /@ cur)

        ConductorField[obs_?MatrixQ, cur_?VectorQ, opt___Rule] :=
         Module[{dr, du, fc, m, normdr},
          fc = FieldComponent /. {opt} /. Options[ConductorField];
          du = Transpose[obs] - cur;
          dr = Transpose@du;
          normdr = #.# & /@ dr;
          m = Reverse /@ dr / normdr;
          {mh, mv} = Transpose[m];
          Which[fc === Vertical, mv,
           fc === Horizontal, mh,
           fc === All, m]
         ]
        ConductorField[obs_?MatrixQ, cur_?MatrixQ, opt___Rule] :=
         Plus @@ (ConductorField[obs, #, opt] & /@ cur)
```

■ Examples of the use of `ConductorField`

□ The use of matrices to determine the current distribution

One conductor

The observation points matrix defines what we call the median plane (as a reference to the real case – a dipole – we started from), and is calculated as follow :

```
x = Range[201] - 101;
y = Table[0, {n, 0, 200}];
ob = Transpose[{x, y}];
```

The current matrix is given as a list :

```
cu = {0, 1};
```

The solution of the system is then given simply using `ConductorField` as follow :

```
Transpose@ConductorField[ob, cu, FieldComponent → All]
```

$$\begin{pmatrix} -\frac{1}{10001} & -\frac{1}{9802} & -\frac{1}{9605} & -\frac{1}{9410} & -\frac{1}{9217} & -\frac{1}{9026} & -\frac{1}{8837} & -\frac{1}{8650} & -\frac{1}{8465} & -\frac{1}{8282} & -\frac{1}{8101} \\ -\frac{100}{10001} & -\frac{99}{9802} & -\frac{98}{9605} & -\frac{97}{9410} & -\frac{96}{9217} & -\frac{95}{9026} & -\frac{94}{8837} & -\frac{93}{8650} & -\frac{92}{8465} & -\frac{91}{8282} & -\frac{90}{8101} \end{pmatrix}$$







Two conductors

The observation matrix is exactly the same as the one used in the case with one conductor (see above). While the current matrix is now defined for two conductors symmetrically located compared to the median plane :

```
In[9]:= cu = {{0, 1}, {0, -1}}
```

$$\begin{pmatrix} 0 & 1 \\ 0 & -1 \end{pmatrix}$$

The vertical field produced by this configuration is then :

```
{bx, by} = Transpose@ConductorField[ob, cu, FieldComponent → All]
```

$$\begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ -\frac{200}{10001} & -\frac{99}{4901} & -\frac{196}{9605} & -\frac{97}{4705} & -\frac{192}{9217} & -\frac{95}{4513} & -\frac{188}{8837} & -\frac{93}{4325} & -\frac{184}{8465} & -\frac{91}{4141} & -\frac{180}{8101} \end{pmatrix}$$



N conductors

In this last case, to define the observation points matrix, the dimensions of the real dipole are used, and is generated as follow:

```
nbobs = 1024;
pasobs = (1.035 - 0.57) / (nbobs - 1);
xo = Table[0.57 + pasobs * i, {i, 0, nbobs - 1}];
yo = Table[0, {i, 0, nbobs - 1}];
MatrixForm[ob = Transpose[{xo, yo}]];
```

The current matrix for conductors located in a plane at distance g compared to the median plane also uses the real dipole dimensions. It is defined as follow :

```
nbcur = 20;
pascur = (1.035 - 0.57) / (nbcur - 1);
xc = Table[0.57 + pascur * i, {i, 0, nbcur - 1}];
g = 0.0175;
yc = Table[g, {i, 0, nbcur - 1}];
MatrixForm[cu = Transpose[{xc, yc}]];
```

The current matrix for conductors in two planes located symmetrically compared to the median plane at distance g and −g (real dipole dimension also used) :

```
nbcur = 40;
pascur = (1.035 - 0.57) / (nbcur / 2 - 1);
xch = Table[0.57 + pascur * i, {i, 0, nbcur / 2 - 1}];
xcb = Table[0.57 + pascur * i, {i, 0,  nbcur / 2 - 1}];
xc = Join[xch, xcb];
g = 0.0175;
ych = Table[g, {i, 0, nbcur / 2 - 1}];
ycb = Table[-g, {i, 0, nbcur / 2 - 1}];
yc = Join[ych, ycb];
MatrixForm[cu = Transpose[{xc, yc}]];
```

The coefficients matrix A is then calculated using `ConductorField` as follow :

```
mat = Transpose[ConductorField[ob, #] & /@ cu];
```

The theoretical vector field is given by experimental parameters from the real dipole :

```
xx = Transpose[ob][[1]];
k = Log[18.5] / Log[1.035 / 0.57]
b = 1.85 (xx / 1.035) ^k;
```

The solution of the system is calculated using the *Mathematica* function `LinearSolve` as follow :

```
mt = Transpose[mat];
a = mt.mat;
bb = mt.b;
x64 = LinearSolve[a, bb]
```



20 conductors in one plane or 40 conductors in two planes (green curve) and the theoretical field (orange curve).

The graph (above) show the solution for both cases : 20 conductors in one plane and 40 conductors in two planes. The difference between these two solutions is the intensity of the current in the conductor : the intensity of the current is divided by two in the case of symmetrical planes.



120 conductors in one plane

Increasing the number of conductors (see graph above), the solution gets very close to the theoretical field, but the current should be divided by two in case of a dipole.

◻ Use of symbolic calculations

This functionality of ConductorField is useful two study the effects the inter conductors distance compared to the gap height.

One conductor

With one conductor,the analytic behaviour of the field shows a maximum and a minimum at an abscissa that depends on the value of the gap.

In[195]:=
```
x = .; n = .; g = .; a = n * g;
f1 = ConductorField[x, {a, g}]
```

Out[196]=

$$\frac{g\,n - x}{g^2 + (g\,n - x)^2}$$

In[197]:=
```
g = 1; Plot[Evaluate[f1 /. n → 0], {x, -10, 10}, PlotRange → All,
 PlotStyle → {Red, Orange, Black, Orange, Red},
 AxesLabel → {"x", "By"}];
```



Let n be zero, if the gap is 1 then the maximum is located at $x = -1$with amplitude 0.5 T, and the minimum at $x = 1$with amplitude –0.5 T (see graph above).

Two conductors

Now consider a pair of conductors at ordinate g and abscissa a (see definition section above). The function remains antisymmetric.

In[208]:=
```
n =.; g =.; f2 = ConductorField[x, {{a, g}, {-a, g}}]
df2 = D[f2, x]
```

Out[208]= $$\frac{-g\,n - x}{g^2 + (-g\,n - x)^2} + \frac{g\,n - x}{g^2 + (g\,n - x)^2}$$

Out[209]= $$-\frac{1}{g^2 + (-g\,n - x)^2} - \frac{1}{g^2 + (g\,n - x)^2} + \frac{2\,(-g\,n - x)^2}{\left(g^2 + (-g\,n - x)^2\right)^2} + \frac{2\,(g\,n - x)^2}{\left(g^2 + (g\,n - x)^2\right)^2}$$

In[210]:=
```
p =.;
num = (Numerator[Factor@df2] / 2) /. (x^p_?EvenQ) → y^(p / 2);
sol = y /. (Simplify@Solve[num == 0, y]); g = 1.;
lsol = Transpose@Chop@Simplify[sol /. n → {.1, .5, 1, 1.5}];
xmin = Sqrt@(Select[#, NonNegative] & /@ lsol)
```

Out[213]= {{1.01489}, {1.31111}, {1.86121, 0}, {2.40615, 0.675622}}

Let ge be equal to 1. When extrema of the magnetic field function are real numbers, two more extrema appear (see calculation above and graph below) with increasing n. All extrema get far from vertical axis with increasing distance between conductors.

In[171]:=
```
g = 1; Plot[Evaluate[f2 /.
    n → {.1, .5, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15}],
 {x, -15, 15}, PlotRange → {{-15, 15}, Automatic},
 PlotStyle → {Black, Red, Green, Blue, Orange, Orange, Orange,
    Orange, Orange, Orange, Orange, Orange, Orange, Orange,
    Orange, Orange, Orange}, AxesLabel → {"x", "By "}];
```



N conductors

Let g be equal to 1. With N conductors, a linearization of the field as the inter conductor distance increases is observed. In the meantime, the intensity of the field decreases.



■ Conclusion

We built a *Mathematica* function that allows, from a theoretical field function, to calculate the current distribution for a set of conductors distributed in a plane, and the field components of the magnetic field produced. It also allows to get the symbolic expression of the field produced by this set of conductors. Many improvements can be imagined for this. We are working on...

We thank François Méot, leader of the French FFAG project called RACCAM, that allowed us to discover this interesting problem.

■ References

[1] Y. Ishi, "Accelerator design and construction for FFAG−KUCA ADSR", presentation, Oct. 15th 2004.